Tutorial 8:
Multiplications

New Instructions Introduced

MPY 

MPYU

MPYA

Overview of Tutorial

Multiplication is one application where digital signal processing devices (DSPs) offer a considerable benefit in performance over microcontrollers (MCUs) and microprocessors (MPUs). For example, on the TMS320C5000, a 16-bit by 16-bit multiply can be executed in a single clock cycle, whereas MCUs and MPUs may take 10 times as many clock cycles. The TMS320C5000 offers a range of multiplication instructions, of which we shall now look at three of the most simple.

Loading the T Register

The majority of multiplication instructions make use of a special register known as the T register (temporary register). The T register can be multiplied by a constant or by the contents of a data memory address. 

The T register is treated as a memory-mapped register, and therefore to store an immediate value in the T register we write:

Example 8-1.

	
	

	STM #33h, T
	; Store the immediate value 33h in the T

; register.


To load a value from a data memory address into the T register we use the instruction LD. There is the restriction when using the instruction LD that the value loaded into the T register must be from a data memory address. Some ways of loading the T register are shown in Example 8-2:

Example 8-2.

	
	

	LD 70h, T
	; Direct addressing. Load the T register with

; the contents of the data memory address

; 70h.

	LD *AR7, T
	; Indirect addressing. Load the T register

; with the contents of the data memory 

; address pointed to by auxiliary register

; AR7.

	LD #33h, T
	; Not allowed. Immediate addressing not

; supported.


Say we wish to load a data memory address with the constant 1234h, then copy this value to the T register. Using direct addressing to load the T register we would write:

Example 8-3.

	
	

	LD #1, DP
	; Set data memory page pointer (DP) to page

; 1 to give access to data memory addresses

; 80h to FFh. 

	ST #1234h, 0h
	; Store the constant 1234h at the data

; memory address 80h + 0h = 80h. Data

; memory address 80h now contains 1234h.  

	LD 0h, T
	; Load the T register with the contents of

; the data memory address 80h + 0h = 80h.

; The T register now contains 1234h.


To carry out the same operation as shown in Example 8-3, but this time using indirect addressing we would write:

Example 8-4.

	
	

	STM #80h, AR3
	; Store address of data memory location

; 80h in AR3. AR3 = 80h.

	LD #1234h, *AR3
	; Store constant 1234h at data memory

; address pointed to by auxiliary

; register AR3. Data memory address 80h

; now contains 1234h. 

	LD *AR3, T
	; Load the T register with the contents

; of the data memory address pointed to

; by the auxiliary register AR3.

; The T register now contains 1234h.


In Example 8-4, we have used AR3 as a pointer to the data memory address, but we could have achieved the same results using any other of the auxiliary registers AR0 to AR7.

Multiplying by a Constant

So far, all we have done is to load the T (temporary) register. In order to carry out the multiplication itself, we have a range of instructions at our disposal. To multiply the T  register by a constant, we can use the instruction MPY (multiply) with immediate data. 

In this context, the instruction MPY means “multiply the contents of the T register by”.

The instruction MPY (multiply) takes two operands. The first operand is the immediate value and lies in the range 8000h to +7FFFh (-32768 to +32767). It is not affected by sign-extension mode. The second operand is the destination, that is, where we wish to store the product. The destination can be either accumulator A or accumulator B.

Example 8-5 shows the instruction MPY being used with three different values:

Example 8-5.

	
	

	STM #10, T
	; Store the value 10 decimal in the T

; register

	MPY #10, A
	; Multiply the contents of the T register

; by the constant 10 decimal and put the

; product into accumulator A. Accumulator A

; now contains 100 decimal (00000064h). 

	MPY #0FFFFh, B
	; Multiply the contents of the T register

; by the constant -10 decimal and put the

; product into accumulator B. Accumulator B

; now contains -100 decimal (FFFFFF9Ch). 

	MPY #0, A
	; Multiply the contents of the T register

; by zero. Accumulator A now contains 0000h.


Note that when using the instruction MPY, no reference is made in the mnemonic to the T register.
Multiplying the T Register by the Contents of a Data Memory Address 

As well as multiplying the T register by an immediate value, we can also multiply the T register by a value stored in data memory: 

Example 8-6.

	
	

	LD #2, DP
	; Set data memory page pointer (DP) to page

; 2. Gain access to data memory addresses

; 100h to 17Fh. 

	MPY 3h, A
	; Multiply the contents of the T register

; by the contents of data memory address

; 100h + 3h = 103h. Put the product into

; accumulator A. 


We can also carry out the same operation as shown in Example 8-6, but using indirect addressing. This time we shall use auxiliary register AR6 as a pointer to a data memory and put the product into accumulator B.

Example 8-7.

	
	

	STM #103h, AR6
	; Store the address of data memory location

; 103h in AR6. AR6 = 103h. 

	MPY *AR6, B
	; Multiply the T register by the contents

; of the data memory address pointed to by

; AR6. The product is put into accumulator

; B. 


Unsigned Multiplication

When using the instruction MPY with immediate data, it should be remembered that the constant is always signed, that is, it can be positive or negative. Whether sign-extension mode is on or off has no effect. Example 8-8 shows an unsuccessful attempt to multiply the contents of the T register by FFFFh (65535).

Example 8-8.

	
	

	MPY #0FFFFh, A
	; The intention was to multiply the

; contents of the T register by 65535.

; The immediate value FFFFh has been

; taken to be the negative number -1,

; rather than the positive number 65535.  


Unsigned Multiplications

To carry out a multiplication of two unsigned numbers, either of which is greater than 7FFFh (32767), we use the instruction MPYU (multiply unsigned). The instruction MPYU (multiply unsigned) takes two operands. The first operand is a data memory address and the second operand is the destination, which can be either accumulator A or accumulator B.   

Example 8-9.

	
	

	LD #4, DP
	; Page 4. Gain access to data memory

; addresses 200h to 27Fh.

	MPYU 70h, A
	; Direct addressing. Multiply the

; unsigned contents of the T register by

; the unsigned contents of data memory

; address 200h + 70h = 270h. The product

; is put into accumulator A. 


We can carry out the same operation as shown in Example 8-9, but using indirect addressing:

Example 8-10.

	
	

	STM #270h, AR4 
	; Store the location of data memory

; address 270h in auxiliary register

; AR4. AR4 = 270h.

	MPYU *AR4, A
	; Multiply the contents of the T

; register by the contents of the data

; memory address pointed to by AR4. The

; product is put into accumulator A. 


The instruction MPYU can only be used with the contents of a data memory location, so the following is not allowed:

Example 8-11.

	
	

	MPYU #0FFFFh, A
	; Illegal instruction. Immediate

; addressing not supported. 


Multiplying Two Variables

We have seen how to use two of the multiply instructions and are now ready to look at a more practical application. Let us multiply the 16-bit variable at data memory address 300h by the 16-bit variable at data memory address 301h. The product will be 32-bit. We will store the low 16 bits of the product at data memory address 302h and the high 16 bits of the product at data memory address 303h.

Using direct addressing we would write:

Example 8-12.

	
	

	LD #6, DP
	; Page 6. Gain access to data memory 

; addresses 300h to 37Fh. 

	LD 0h, T
	; Load the T register with the value

; contained at data memory address

; 300h + 0h = 300h.

	MPY 1h, A
	; Multiply the T register by the contents of

; the data memory address 300h + 1h = 301h.

; The product is stored in accumulator A.

	STL A, 2h
	; Store the low word of the product at the

; data memory address 300h + 2h = 302h.

	STH A, 3h
	; Store the high word of the product at the

; data memory address 300h + 3h = 303h.  


We shall now repeat the same operation as carried out in Example 8-12, this time using indirect addressing and accumulator B as the destination.

Example 8-13.

	
	

	STM #300h, AR5
	; Store the address of data memory location

; 300h in auxiliary register AR5. 

; AR5 = 300h. 

	LD *AR5+, T
	; Load the T register with the contents of

; the data memory address pointed to by

; AR5. Increment AR5 to point to the next

; data memory address (301h).  

	MPY *AR5+, B
	; Multiply the T register by the contents

; of the data memory pointed to by

; AR5. The product is stored in accumulator

; B. Increment AR5 to point to the next

; data memory address (302h).

	STL B, *AR5+
	; Store the low word of the product at

; data memory address 302h. Increment AR5

; to point to the next data memory address

; (303h).

	STH B, *AR5
	; Store the high word of the product at the

; data memory address pointed to by AR5.

; No need to increment pointer.   


In Example 8-13, we have used the operand *AR5+ to increment the pointer each time to point to the next higher address. 

There are one or two differences between direct and indirect addressing that the programmer should be aware of. When the data pointer (DP) is already set to the correct page, then direct addressing uses one less instruction. However, indirect addressing is preferred when there is a series of multiplications. This is because it is possible to increment the auxiliary register to point to the next data memory address.  

Combining Instructions

Consider the code shown in Example 8-14 that is used to multiply the contents of the T register by an immediate value:

Example 8-14.

	
	

	LD #3, DP
	; Page 3. Gain access to data memory

; addresses 180h to 1FFh.

	LD 2h, T
	; Load the T register with the value

; contained at data memory address 180h +

; 2h = 182h.

	MPY #492h, A
	; Multiply the contents of the T register

; by the constant 492h. The product is

; stored in accumulator A.


We can in fact combine the operation to load the T register and carry out a multiplication into a single instruction. For this we use the instruction MPY (multiply) with three operands. The first operand is the source data memory address to be multiplied. The second operand is the constant and the third operand is the destination which can be accumulator A or accumulator B. Example 8-15 shows the usage of the instruction MPY with three operands:   

Example 8-15.

	
	

	LD #3, DP
	; Page 3. Gain access to data memory

; addresses 180h to 1FFh. 

	MPY 2h, #492h, A
	; Load the T register with the value

; contained at data memory 

; address 180h + 2h = 182h. Multiply

; the contents of the T register by the

; immediate value 492h and put the

; product into accumulator A.


In this case the operation of transferring the contents of the data memory address to the T register is done automatically as part of the instruction MPY. 

We can carry out the same operation as shown in Example 8-15, but this time using indirect addressing and accumulator B as the destination.  

Example 8-16.

	
	

	STM #182h, AR2
	; Store the address of data memory

; location 182h in auxiliary register

; AR2. AR2 = 182h. 

	MPY *AR2, #492h, B
	; Load the T register with the value

; at the address pointed to by

; auxiliary register AR2. Multiply

; the contents of the T register by the

; immediate value 492h and put the

; product into accumulator B.


In Example 8-15, the use of the T register has become totally transparent to the user. As far as the user is concerned, the contents of the first operand are being multiplied by an immediate value and the result is placed in the destination. 

In practical terms, a multiplication using three operands is the most efficient way to use the instruction MPY.

Multiplication using an Accumulator

This variation of the multiply instruction allows us to multiply the T register by a value in accumulator A. For this we use the instruction MPYA (multiply by accumulator A).  

This instruction is particularly useful when we carry out a series of multiplications, for example for a polynomial expression.

Example 8-17.

	
	

	MPYA B
	; Multiply the contents of the T register

; by the high word of accumulator A. Put

; the product into accumulator B.   


The instruction MPYA (multiply by accumulator A) takes one operand only, and that is the destination of accumulator A or B. We are also restricted to working with accumulator A only as the source, and cannot multiply the contents of the T register by accumulator B.

An easy mistake to make would be to expect the instruction MPYA to work on the low word of accumulator A. This is not the case. The instruction MPYA (multiply by accumulator A), multiplies the T register by the high word of accumulator A. 

To load accumulator A with a 16-bit constant before carrying out a multiplication, we must use the load instruction LD with a shift of 16. 

Example 8-18.

	
	

	STM #2, T
	; Store the value 2 in the T register.

	LD #300h,16, A
	; Load accumulator A with the value 300h

; shifted 16 places to the left. 

; Accumulator A now contains 03000000h.    

	MPYA B
	; Multiply the T register by the high

; word of accumulator A. Put the product 

; into accumulator B. Accumulator B now

; contains 00000600h.     


Example 8-19 shows an unsuccessful attempt to multiply the 16-bit constant in accumulator A by the contents of the T register:

Example 8-19.

	
	

	STM #2, T
	; Store the immediate value 2 in the T

; register.

	LD #300h,A
	; Load accumulator A with the value 300h.

; Accumulator A now contains 00000300h.    

	MPYA B
	; Multiply the T register by the high

; word of accumulator A. Put product 

; into accumulator B. Accumulator B now

; contains 0h.     


Omitting the shift when loading accumulator A prior to using the instruction MPYA will cause the incorrect product to be generated.

The instruction MPYA is not affected by sign-extension mode.

Upgrading from the TMS320C2000 to the TMS320C5000

On the TMS320C5000, the use of the T register is less important for multiplications than with the TMS320C2000, which has to use the T register to carry out multiplications. 
The multiplication instructions on the TMS320C5000 do differ considerably from the TMS320C2000. Details are given in Table 8-1.   

Table 8-1: Comparison of Instructions

	Description
	TMS320C2000  Instruction
	TMS320C5000 

Instruction

	Load T register
	LT *
	LD *ARx, T

	Store immediate value in T register
	Not supported
	STM #40h, T

	Multiply T register by constant
	MPY #30
	MPY #30, Acc or

MPY #30, Acc

	Multiply T register using direct addressing
	MPY 80h
	MPY 80h, Acc

or

MPY 80h, Acc

	Multiply T register using indirect addressing
	MPY *
	MPY *ARx, Acc

or

MPY *ARx, Acc

	Multiply dmad by constant and store in accumulator 
	Not 

supported
	MPY *ARx, #300h, Acc

	Transfer P register to accumulator
	PAC
	Not used

	Transfer high word of P register to data memory.


	SPH 80h
	Not used

	Transfer low word of P register to data memory .
	SPL 81h
	Not used

	Multiply high word of accumulator by T register
	Not supported
	MPYA Acc


ARx = An auxiliary register AR0 to AR7. Acc = Accumulator A or Accumulator B.

The TMS320C2000 places the product in the P register (product register). This means that the product has to be transferred to either the accumulator or a data memory location using the instructions PAC, SPH and SPL. The TMS320C5000 no longer has a P register (32-bit product), being able to place the product directly in accumulator A or accumulator B. This means the instructions PAC, SPH and SPL are no longer supported.

On the TMS320C5000, an improvement has been made to multiplication by an immediate value. On the TMS320C2000, this operation is restricted to the use of a 13-bit constant. This meant it is only possible to multiply by a value in the range 2000h to 1FFF  (-4096 to +4095). The TMS320C5000 supports 16-bit constants. It is possible to multiply by a constant in the range 8000h to 7FFFh (-32768 to +32767). 

A further improvement offered by the TMS320C5000 is that an immediate value can be stored directly in the T register.

A point of potential confusion is that on the TMS320C5000, the instruction MPYA means multiply by accumulator A. However, on the TMS320C2000, the instruction MPYA meant multiply with accumulate. The two are quite different functions. The mnemonic MPYA should therefore be treated with some caution.

Questions

	1.
	What does the letter T in T register stand for?

a) 
target 

b) 
temporary 

c) 
tertiary

d) 
timing. 

	2.
	Which three of the following TMS320C5000 instructions are legal?

a) 
STM #45h, T

b) 
LD #22h, T

c) 
LD 5h, T

c) 
LD *AR4, T

d) 
LD *AR1+, T 

	3.
	How do we clear the T register to zero ?

	4.
	When would we use the instruction MPYU in preference to the instruction MPY?

	5.
	Which three of the following instructions are legal?

a) 
MPYU #100, A

b) 
MPYU *AR3, B

c) 
MPYU 70h, A

d)  MPYU *AR6-, A  

	6.
	How do we multiply two 16-bit variables together?

	7.
	Why might we use indirect addressing to multiply a series of values rather than direct addressing?

	8.
	Reduce the following two instructions to a single instruction:

LD 0h, T

MPY #10h, B

	9.
	Why do we not use the instruction MPYA to multiply accumulator B?

	10.
	When loading a variable with the instruction MPYA , why must we use a shift?

